

Интегрированные математические пакеты


Юдинцев В. В.

Самарский университет
27.02.2021

Лекция 2

Построение модели движения системы материальных точек, связанных пружинами. Использование элементов функционального программирования.

Схема



Количество тел

In[1]:= $n = 4$;

Относительные координаты материальных точек

```
In[2]:= q = Map[δ#[t] &, Range[n]]
```

```
q = δ#[t] & /@ Range[n]
```

```
dq = D[q, t]
```

```
Out[2]= {δ1[t], δ2[t], δ3[t], δ4[t]}
```

```
Out[3]= {δ1[t], δ2[t], δ3[t], δ4[t]}
```

```
Out[4]= {δ1'[t], δ2'[t], δ3'[t], δ4'[t]}
```

Абсолютные координаты материальных точек

```
In[5]:= x = Accumulate[q]
```

```
Out[5]= {δ1[t], δ1[t] + δ2[t], δ1[t] + δ2[t] + δ3[t], δ1[t] + δ2[t] + δ3[t] + δ4[t]}
```

Левая часть дифференциальных уравнений

```
In[6]:= lSide = m D[x, {t, 2}]
```

```
Out[6]= {m δ1''[t], m (δ1''[t] + δ2''[t]),  
m (δ1''[t] + δ2''[t] + δ3''[t]), m (δ1''[t] + δ2''[t] + δ3''[t] + δ4''[t])}
```

Список сил

```
In[7]:= F = Map[(# - l0) c &, q]
```

```
Out[7]= {c (-l0 + δ1[t]), c (-l0 + δ2[t]), c (-l0 + δ3[t]), c (-l0 + δ4[t])}
```

Правая часть дифференциальных уравнений (силы)

Использование функции Apply

```
In[8]:= Fi = Map[-#[[1]] + #[[2]] &, Append[Partition[F, 2, 1], {F[[-1]], 0}]]
```

```
Out[8]= {-c (-l0 + δ1[t]) + c (-l0 + δ2[t]), -c (-l0 + δ2[t]) + c (-l0 + δ3[t]),  
-c (-l0 + δ3[t]) + c (-l0 + δ4[t]), -c (-l0 + δ4[t])}
```

Использование функции Apply (постфиксная запись)

```
In[9]:= Fi = -#[[1]] + #[[2]] & /@ Append[Partition[F, 2, 1], {F[[-1]], 0}]
```

```
Out[9]= {-c (-l0 + δ1[t]) + c (-l0 + δ2[t]), -c (-l0 + δ2[t]) + c (-l0 + δ3[t]),  
-c (-l0 + δ3[t]) + c (-l0 + δ4[t]), -c (-l0 + δ4[t])}
```

Уравнения движения

Использование функции Apply

```
In[10]:= #[[1]] == #[[2]] & /@ Transpose[{lSide, Fi}]
Out[10]= {m δ1''[t] == -c (-l0 + δ1[t]) + c (-l0 + δ2[t]),
          m (δ1''[t] + δ2''[t]) == -c (-l0 + δ2[t]) + c (-l0 + δ3[t]),
          m (δ1''[t] + δ2''[t] + δ3''[t]) == -c (-l0 + δ3[t]) + c (-l0 + δ4[t]),
          m (δ1''[t] + δ2''[t] + δ3''[t] + δ4''[t]) == -c (-l0 + δ4[t])}
```

Использование функции MapThread

```
In[11]:= eq = MapThread[#1 == #2 &, {lSide, Fi}]
Out[11]= {m δ1''[t] == -c (-l0 + δ1[t]) + c (-l0 + δ2[t]),
          m (δ1''[t] + δ2''[t]) == -c (-l0 + δ2[t]) + c (-l0 + δ3[t]),
          m (δ1''[t] + δ2''[t] + δ3''[t]) == -c (-l0 + δ3[t]) + c (-l0 + δ4[t]),
          m (δ1''[t] + δ2''[t] + δ3''[t] + δ4''[t]) == -c (-l0 + δ4[t])}
```

Начальные условия

```
In[12]:= nu = {
  MapThread[#1 == #2 &, {q /. t → 0, Append[ConstantArray[l0, n - 1], l0 + δ0]}],
  MapThread[#1 == #2 &, {D[q, t] /. t → 0, ConstantArray[0, n]}]
} // Flatten
Out[12]= {δ1[0] == l0, δ2[0] == l0, δ3[0] == l0,
          δ4[0] == l0 + δ0, δ1'[0] == 0, δ2'[0] == 0, δ3'[0] == 0, δ4'[0] == 0}
```

Параметры системы

```
In[13]:= params = {m → 1.0, c → 50, l0 → 1.0, δ0 → 0.5}
Out[13]= {m → 1., c → 50, l0 → 1., δ0 → 0.5}


In[14]:= {a, b, c} /. {a → b + 1, b → 3, c → 4}
          {a, b, c} //. {a → b + 1, b → 3, c → 4}
Out[14]= {1 + b, 3, 4}
Out[15]= {4, 3, 4}
```

Интегрирование дифференциальных уравнений


```
In[16]:= tk = 5;
sol = NDSolve[{eq, nu} /. params, {q, dq} // Flatten, {t, 0, tk}] // Flatten
```

```
Out[17]= { $\delta_1[t]$  → InterpolatingFunction[ Domain: {{0., 5.}} Output: scalar] [t],
 $\delta_2[t]$  → InterpolatingFunction[ Domain: {{0., 5.}} Output: scalar] [t],
 $\delta_3[t]$  → InterpolatingFunction[ Domain: {{0., 5.}} Output: scalar] [t],
 $\delta_4[t]$  → InterpolatingFunction[ Domain: {{0., 5.}} Output: scalar] [t],
 $\delta_1'[t]$  → InterpolatingFunction[ Domain: {{0., 5.}} Output: scalar] [t],
 $\delta_2'[t]$  → InterpolatingFunction[ Domain: {{0., 5.}} Output: scalar] [t],
 $\delta_3'[t]$  → InterpolatingFunction[ Domain: {{0., 5.}} Output: scalar] [t],
 $\delta_4'[t]$  → InterpolatingFunction[ Domain: {{0., 5.}} Output: scalar] [t]}
```

```
In[18]:= Plot[q /. sol // Evaluate, {t, 0, tk},
PlotLegends → (q /. s_[t] → s), PlotTheme → {"Presentation", "FrameGrid"}]
```



```
In[19]:= Plot[x /. sol // Evaluate, {t, 0, tk},
  PlotLegends -> Map["x"# &, Range[n]], PlotTheme -> {"Presentation", "FrameGrid"}]
```



Верификация модели


```
In[20]:= Ek = Total[ $\frac{m D[x, t]^2}{2}$ ]
```

```
Ep = Total[ $\frac{c (q - l_0)^2}{2}$ ]
```

Out[20]= $\frac{1}{2} m \delta_1' [t]^2 + \frac{1}{2} m (\delta_1' [t] + \delta_2' [t])^2 +$
 $\frac{1}{2} m (\delta_1' [t] + \delta_2' [t] + \delta_3' [t])^2 + \frac{1}{2} m (\delta_1' [t] + \delta_2' [t] + \delta_3' [t] + \delta_4' [t])^2$

Out[21]= $\frac{1}{2} c (-l_0 + \delta_1 [t])^2 + \frac{1}{2} c (-l_0 + \delta_2 [t])^2 + \frac{1}{2} c (-l_0 + \delta_3 [t])^2 + \frac{1}{2} c (-l_0 + \delta_4 [t])^2$

```
In[22]:= Plot[{Ep, Ek, Ek + Ep} /. params /. sol // Evaluate, {t, 0, tk},
  PlotLegends -> {"П", "Т", "Т+П"}, PlotTheme -> {"Presentation", "FrameGrid"}]
```



Анимация

Функция, которая рисует пружину (зигзаг)

x1 начало пружины


x2 конец пружины

d диаметр пружины

c количество “витков”

y координата y

```
In[23]:= Spring[x1_, x2_, d_, c_, y_] :=
  Line[Table[{x1 +  $\frac{x2 - x1}{2 c} i$ , y + (-1)^i d * 0.5}, {i, 0, 2 c}]];
Graphics[Spring[0, 1, 0.5, 10, 0]]
```



Определение точек начала и конца пружины

```
In[25]:= Prepend[x, 0]
```

```
Partition[%, 2, 1]
```

```
Out[25]= {0,  $\delta_1[t]$ ,  $\delta_1[t] + \delta_2[t]$ ,  $\delta_1[t] + \delta_2[t] + \delta_3[t]$ ,  $\delta_1[t] + \delta_2[t] + \delta_3[t] + \delta_4[t]$ }
```

```
Out[26]= {{0,  $\delta_1[t]$ }, { $\delta_1[t]$ ,  $\delta_1[t] + \delta_2[t]$ }, { $\delta_1[t] + \delta_2[t]$ ,  $\delta_1[t] + \delta_2[t] + \delta_3[t]$ },
  { $\delta_1[t] + \delta_2[t] + \delta_3[t]$ ,  $\delta_1[t] + \delta_2[t] + \delta_3[t] + \delta_4[t]$ }}
```

```

In[27]:= Manipulate[
  Graphics[
    {
      Darker[Green, 0.3],
      Spring[#[[1]], #[[2]], 0.05, 20, 0] & /@ Partition[Prepend[x, 0], 2, 1] /. sol /.
      t → tf,
      Red,
      Map[Disk[{-#, 0}, 0.1] &, x /. sol /. t → tf]
    },
    Frame -> True, PlotRange -> {{0, l0 (n + 1) /. params}, {-1, 1}}
  ], {tf, 0, tk}]

```

Out[27]=

